ISR: Lecture 1

José Meseguer

Computer Science Department
University of Illinois at Urbana-Champaign

Equational Theories

Theories in equational logic are called equational theories. In Computer Science they are sometimes referred to as algebraic specifications.

An equational theory is a pair (Σ, E), where:

- Σ, called the signature, describes the syntax of the theory, that is, what types of data and what operation symbols (function symbols) are involved;

- E is a set of equations between expressions (called terms) in the syntax of Σ.

Unsorted, Many-Sorted, and Order-Sorted Signatures

Our syntax Σ can be more or less expressive, depending on how many types (called sorts) of data it allows, and what relationships between types it supports:

- unsorted (or single-sorted) signatures have only one sort, and operation symbols on it;
- many-sorted signatures allow different sorts, such as `Integer`, `Bool`, `List`, etc., and operation symbols relating these sorts;
- order-sorted signatures are many-sorted signatures that, in addition, allow inclusion relations between sorts, such as `Natural < Integer`.

Maude Functional Modules

Maude functional modules are equational theories (Σ, E), declared with syntax

$$\texttt{fmod}\ (\Sigma, E)\ \texttt{endfm}$$

Such theories can be unsorted, many-sorted, or order-sorted.

In what follows we will see examples of unsorted, many-sorted and order-sorted equational theories (Σ, E) expressed as Maude functional modules, and of how one can use such theories as functional programs by computing with the equations E.

Unsorted Functional Modules

```
*** prefix syntax

fmod NAT-PREFIX is
  sort Natural .
  op 0 : -> Natural [ctor] .
  op s : Natural -> Natural [ctor] .
  op plus : Natural Natural -> Natural .
  vars N M : Natural .
  eq plus(N,0) = N .
  eq plus(N,s(M)) = s(plus(N,M)) .
endfm

Maude> red plus(s(s(0)),s(s(0))) .
reduce in NAT-PREFIX : plus(s(s(0)), s(s(0))) .
rewrites: 3 in -10ms cpu (0ms real) (~ rewrites/second)
result Natural: s(s(s(s(0))))
Maude>
```

Unsorted Functional Modules (II)

```
fmod NAT-MIXFIX is                      *** mixfix syntax
  sort Natural .
  op 0 : -> Natural [ctor] .
  op s : Natural -> Natural [ctor] .
  op _+_ : Natural Natural -> Natural .
  op _*_ : Natural Natural -> Natural .
  vars N M : Natural .
  eq N + 0 = N .
  eq N + s(M) = s(N + M) .
  eq N * 0 = 0 .
  eq N * s(M) = N + (N * M) .
endfm

Maude> red s(s(0)) + s(s(0)) .
reduce in NAT-MIXFIX : s(s(0)) + s(s(0)) .
rewrites: 3 in 0ms cpu (0ms real) (~ rewrites/second)
result Natural: s(s(s(s(0))))
Maude>
```

Many-Sorted Functional Modules

```
fmod NAT-LIST is
  protecting NAT-MIXFIX .
  sort List .
  op nil : -> List [ctor] .
  op _;_ : Natural List -> List [ctor] .
  op length : List -> Natural .
  var N : Natural .
  var L : List .
  eq length(nil) = 0 .
  eq length(N ; L) = s length(L) .
endfm

Maude> red length(0 ; (s(0) ; (s(s(0)) ; (0 ; nil)))) .
reduce in NAT-LIST : length(0 ; s 0 ; s s 0 ; 0 ; nil) .
rewrites: 5 in 0ms cpu (0ms real) (~ rewrites/second)
result Natural: s(s(s(s(0))))
Maude>
```

Many-Sorted Signatures

The many-sorted signature Σ of the NAT-LIST example is:

```
sorts Natural List .
op 0 : -> Natural .
op s : Natural -> Natural .
op _+_ : Natural Natural -> Natural .
op _*_ : Natural Natural -> Natural .
op nil : -> List .
op _;_ : Natural List -> List .
op length : List -> Natural .
```

In general, a many-sorted signature Σ is a pair $\Sigma = (S, F)$, with S the set of sorts, and F a set of function declarations of the form $f : s_1 \ldots s_n \to s$, with $s_1, \ldots, s_n, s \in S$ and $n \geq 0$.

In Maude, the set of sorts S is declared with the `sorts` keyword, and the function declarations F with the `op` keyword.

The Need for Order-Sorted Signatures

Many-sorted signatures are still too restrictive. The problem is that some operations are partial, and there is no natural way of defining them within a many-sorted framework.

Consider for example defining a function `first` that takes the first element of a list of natural numbers, or a predecessor function `p` that assigns to each natural number its predecessor. What can we do? If we define,

```
op first : List -> Natural .
op p : Natural -> Natural .
```

we have then the awkward problem of defining the values of `first(nil)` and of `p(0)`, which in fact are undefined.

The Need for Order-Sorted Signatures (II)

A much better solution is to recognize that these functions are partial with the typing just given, but become total on appropriate subsorts `NeList < List` of nonempty lists, and `NzNatural < Natural` of nonzero natural numbers. If we define,

```
op s : Natural -> NzNatural .
op _;_ : Natural List -> NeList .
op first : NeList -> Natural .
op p : NzNatural -> Natural .
```

everything is fine. Subsorts also allow us to overload operator symbols. For example, `Natural < Integer`, and

```
op _+_ : Natural Natural -> Natural .
op _+_ : Integer Integer -> Integer .
```

Order-Sorted Functional Modules

```
fmod NATURAL is
  sorts Natural NzNatural .
  subsorts NzNatural < Natural .
  op 0 : -> Natural [ctor] .
  op s : Natural -> NzNatural [ctor] .
  op p : NzNatural -> Natural .
  op _+_ : Natural Natural -> Natural .
  op _+_ : NzNatural NzNatural -> NzNatural .
  vars N M : Natural .
  eq p(s(N)) = N .
  eq N + 0 = N .
  eq N + s(M) = s(N + M) .
endfm

Maude> red p(s(s(0)) + s(s(0))) .
reduce in NATURAL : p(s(s(0)) + s(s(0))) .
rewrites: 4 in 0ms cpu (0ms real) (~ rewrites/second)
result NzNatural: s(s(s(0)))
```

Order-Sorted Functional Modules (II)

```
fmod NAT-LIST-II is
  protecting NATURAL .
  sorts NeList List .
  subsorts NeList < List .
  op nil : -> List [ctor] .
  op _;_ : Natural List -> NeList [ctor] .
  op length : List -> Natural .
  op first : NeList -> Natural .
  op rest : NeList -> List .
  var N : Natural .
  var L : List .
  eq length(nil) = 0 .
  eq length(N ; L) = s length(L) .
  eq first(N ; L) = N .
  eq rest(N ; L) = L .
endfm
```

Order-Sorted Signatures

An order-sorted signature Σ is a pair $\Sigma = ((S, <), F)$ where (S, F) is a many-sorted signature, and where $<$ is a partial order relation on the set S of sorts called subsort inclusion.

That is, $<$ is a binary relation on S that is:

- *irreflexive*: $\neg(x < x)$
- *transitive*: $x < y$ and $y < z$ imply $x < z$

Any such relation $<$ has an associated $\leq$ relation that is *reflexive*, *antisymmetric*, and *transitive*. We will move back and forth between $<$ and $\leq$ (see *STACS* 7.4).

Note: Unless specified otherwise, by a signature we will always mean an order-sorted signature.

Connected Components of the Poset of Sorts

Given a signature Σ, we can define an equivalence relation (see *STACS* 7.6) $\equiv_{\leq}$ between sorts $s, s' \in S$ as the transitive closure $(\leq \cup \geq)^{+}$, that is, as the smallest relation such that:

- if $s \leq s'$ or $s' \leq s$ then $s \equiv_{\leq} s'$
- if $s \equiv_{\leq} s'$ and $s' \equiv_{\leq} s''$ then $s \equiv_{\leq} s''$

We call the equivalence classes modulo $\equiv_{\leq}$ the connected components of the poset order $(S, \leq)$. Intuitively, when we view the poset as a directed acyclic graph, they are the connected components of the graph (*STACS* 7.6, Exercise 68).

Notation: For each $s \in S$ its connected component $[s]_{\equiv_{\leq}}$ is abbreviated to $[s]$.

Connected Components Example



$$S/\equiv_{\leq} = \{\{NzNatural, Natural, NzInteger, Integer\}, \{Nelist, List\}, \{Bool, Prop\}\}$$

Subsort vs. Ad-hoc Overloading

In general, the same operator name may have different declarations in the same signature Σ. For example, in the `NATURAL` module we have,

```
op _+_ : Natural Natural -> Natural .
op _+_ : NzNatural NzNatural -> NzNatural .
```

When we have two operator declarations, $f : w \longrightarrow s$, and $f : w' \longrightarrow s'$, with w and w' strings of equal length, then: (1) if $w \equiv_{\leq} w'$ and $s \equiv_{\leq} s'$, we call them subsort overloaded; (2) otherwise, e.g, _+_ for `Natural` and for exclusive or in `Bool`, we call them ad-hoc overloaded.

The Kind-Completion of a Signature

Were do error terms like $p(0)$ or $head(nil)$ live? Actually nowhere, unless we extend the given order-sorted signature $\Sigma = ((S, \leq), F)$ to its so-called kind-completion $\widehat{\Sigma}$ by adding:

- for each connected component $[s]$ a fresh new sort $\top_{[s]}$ (called the kind of $[s]$), as top element of the component, i.e., $\forall s' \in [s]\ s' < \top_{[s]}$
- for each operator declaration $f : s_1 \ldots s_n \to s$ in F a new declaration $f : \top_{[s_1]} \ldots \top_{[s_n]} \to \top_{[s]}$.

Now, all error terms that do not have a sort in Σ but nevertheless make sense can be typed as having a kind $\top_{[s]}$ in $\widehat{\Sigma}$. Maude performs the completion $\Sigma \mapsto \widehat{\Sigma}$ automatically.

Exercises

Ex.1.1. Define in Maude the following functions on the naturals:

- $>$ and $\geq$ as Boolean-valued binary functions importing the built-in module `BOOL` with single sort `Bool`.
- **max** and **min**, that yield the maximum, resp. minimum, of two numbers,
- **even** and **odd** as Boolean-valued functions on the naturals,
- **factorial**, the factorial function.

Exercises (II)

Ex.1.2. Define in Maude the following functions on list of natural numbers:

- **append** and **reverse**, which appends two lists, resp. reverses the list,

- **max** and **min** that computes the biggest (resp. smallest) number in the list,

- **get.even**, which extracts the lists of even numbers of a list,